

## Dodatek A — Instrukcje i deklaracje.

Poniżej zestawione zostały w kolejności alfabetycznej instrukcje i deklaracje dostępne w programie PSpice. Każda z nich uzupełniona jest o przykład użycia i komentarz.

---

### \* KOMENTARZ

Postać ogólna:

\* [komentarz]

Przykład:

\* to jest przykład

Linia zaczynająca się od gwiazdki „ \* ” nie wpływa na działanie programu PSpice. Zaleca się dokładne komentowanie opisu analizowanego obwodu tak, jak gdyby użytkownikowi groził w najbliższym czasie zanik pamięci. Dobrą praktyką jest opisywanie węzłów deklarowanego podobwodu.

Przykład:

```
*wzmacniacz operacyjny
*           we+
*           |   we-
*           |   |   wy
*           |   |   |
.SUBCKT UA741 1   2   10
```

Patrz strona: 5.

---

### ; KOMENTARZ W LINII OPISU

Postać ogólna:

[linia\_opisu];[komentarz]

Przykład:

```
R13 6 8 10K ;rezystor sprzężenia zwrotnego
C1 10 0 0.1U ;pojemność pasożytnicza
```

Średnik „ ; ” traktowany jest przez program PSpice tak samo jak znak końca linii. Po natrafieniu na średnik program przechodzi do interpretacji następnej linii. W ten sposób na końcu linii opisu obwodu, za średnikiem można umieścić różne użyteczne uwagi.

Patrz strona: 3.

**.AC****ANALIZA ZMIENNOPRĄDOWA**

Postać ogólna:

```
.AC [LIN][OCT][DEC] l_punktów _start _stop
```

Przykłady:

```
.AC LIN 101 100HZ 200HZ
.AC OCT 10 1KHZ 16KHZ
.AC DEC 20 1MEG 100MEG
```

Analiza wykonywana jest dla częstotliwości źródła (źródeł) prądu zmiennego zmieniającej się od wartości **\_start** do wartości **\_stop**. Słowa kluczowe **LIN**, **OCT** i **DEC** określają sposób w jaki zmieniana jest częstotliwość.

**LIN** Zmiany częstotliwości zachodzą liniowo. W zakresie częstotliwości od **\_start** do **\_stop** uwzględniona zostanie liczba punktów określona przez parametr **l\_punktów**.

**OCT** Zmiany częstotliwości zachodzą oktawami. W każdej oktawie uwzględniona zostanie liczba punktów określona przez parametr **l\_punktów**.

**DEC** Zmiany częstotliwości zachodzą dekadami. W każdej dekadzie uwzględniona zostanie liczba punktów określona przez parametr **l\_punktów**.

Patrz strona: 45.

**.DC****ANALIZA STAŁOPRĄDOWA**

Postać ogólna:

```
.DC [LIN] _nazwa_wielkości _start _stop _krok
+ [zagnieżdżona_specyfikacja_zmian]
.DC [OCT][DEC] _nazwa_wielkości _start _stop l_punktów
+ [zagnieżdżona_specyfikacja_zmian]
.DC _nazwa_wielkości [LIST] <lista_wartości>
+ [zagnieżdżona_specyfikacja_zmian]
```

Przykłady:

```
.DC VIN -0.25 0.25 0.05
.DC LIN 12 5mA -2mA 0.1mA
.DC VCE 0V 10V 0.5V IB 0mA 1mA 50uA
.DC RES RMOD(R) 0.9 1.1 0.001
.DC DEC NPN QFASR(IS) 1e-18 1e-14 5
.DC TEMP LIST 0 20 27 50 80 100 -50
```

Wyznaczane są charakterystyki stałoprądowe. W polu **\_nazwa\_wielkości** umieszcza się nazwę wielkości, która będzie zmieniana. Może nią być:

- Nazwa niezależnego źródła napięcia lub prądu.
- Parametr modelu. W tym wypadku w polu **\_nazwa\_wielkości** podaje się:
  - typ modelu;
  - nazwę modelu;
  - parametr modelu podany w nawiasach.

Przykład czwarty powyżej dotyczy zmian parametru modelu oporności (typ modelu RES). Model nazywa się RMOD, natomiast zmieniana będzie wartość parametru R.

Kolejny przykład dotyczy modelu tranzystora bipolarnego n–p–n o nazwie QFASR. Zmieniana będzie wartość parametru IS.

- Temperatura otoczenia analizowanego układu. Identyfikowana jest przez słowo kluczowe **TEMP** — ostatni przykład powyżej.

Sposób zmian określony jest przez podanie jednego z następujących słów kluczowych:

**LIN** Wartość wielkości zmieniana jest liniowo od wartości **\_start** do wartości **\_stop** z krokiem **\_krok**.

**DEC** Wartość wielkości zmieniana jest logarytmicznie, dekadami od wartości **\_start** do wartości **\_stop**. Liczba punktów przypadająca na każdą dekadę określona jest przez parametr **I\_punktów**.

**OCT** Wartość wielkości zmieniana jest logarytmicznie, oktawami od wartości **\_start** do wartości **\_stop**. Liczba punktów przypadająca na każdą oktawę określona jest przez parametr **I\_punktów**.

**LIST** Wartości zmienianej wielkości podane są na liście **<\_lista\_wartości>**.

Dopuszczalne jest zagnieżdżenie zmian tzn. podanie następczej specyfikacji wielkości, która będzie zmieniana. W tym przypadku dla każdej wartości pierwszej z podanych wielkości zostaną wykonane obliczenia dla wszystkich wartości drugiej wielkości.

Patrz strony: 20, 35, 136.

## **.END**

## **KONIEC OBWODU**

Postać ogólna:

**.END**

Przykład:

**.END**

Deklaracja **.END** oznacza koniec obwodu. Opis obwodu, wszystkie instrukcje i deklaracje powinny znaleźć się przed deklaracją **.END**. W jednym zbiorze dyskowym może znaleźć się wiele deklaracji **.END** oddzielających *niezależne* zadania — opis obwodu+instrukcje i deklaracje programu PSpice. Program po zakończeniu jednego zadania *zeruje wszystkie zmienne* i przechodzi do kolejnego zadania. Ułatwia to zorganizowanie ciągu obliczeń, który potrwa np. całą noc.

Patrz strony: 3, 5, 22.

## **.ENDS KONIEC DEKLARACJI STRUKTURY PODOBWODU**

Postać ogólna:

**.ENDS** [\_nazwa\_podobwodu]

Przykłady:

**.ENDS**

**.ENDS WZMOP**

Deklaracja końca definicji struktury podobwodu. Jeśli podana zostanie bez nazwy podobwodu (pole **\_nazwa\_podobwodu**) oznacza to koniec definicji struktury wszystkich „otwartych” podobwodów. Zaleca się podawanie nazwy podobwodu, którego deklarację chcemy zakończyć.

Patrz strona: 68.

---

**.FOUR** ANALIZA FOURIEROWSKA

Postać ogólna:

```
.FOUR _częstotliwość <_lista_wielkości>
```

Przykład:

```
.FOUR 10kHz V(5) V(6,7) I(V2)
```

Przebieg czasowy każdej z wielkości wyszczególnionych na liście <\_lista\_wielkości> rozkładany jest w trygonometryczny szereg Fourier-a. Częstotliwość podstawową podaje się w polu **\_częstotliwość**. Analiza Fourierowska musi być poprzedzona analizą w dziedzinie czasu. Rozkład przebiegu dokonywany jest w przedziale czasu o długości równej odwrotności częstotliwości składowej podstawowej. Przedział ten umieszczony jest na końcu przedziału czasu, w którym wykonywana jest analiza stanów nieustalonych.

Patrz strona: 94.

---

**.IC** WARUNKI POCZĄTKOWE

Postać ogólna:

```
.IC V(_węzeł1)=_w1 [V(_węzeł2)=_w2 ...]
```

Przykład:

```
.IC V(1)=1V V(2)=1.5V V(3)=3V
```

Podczas obliczania statycznego punktu pracy *przed analizą stanów nieustalonych* każda z wartości **\_w1, \_w2, ...** przypisana jest potencjałowi węzła o numerze **\_węzeł1, \_węzeł2, ...** Po obliczeniu statycznego punktu pracy z uwzględnieniem ustalonych, za pomocą instrukcji **.IC**, wartości potencjałów analiza stanów nieustalonych przebiega dalej normalnie. Jeżeli w instrukcji analizy stanów nieustalonych **.TRAN** użyto słowa kluczowego **UIC** to wszystkie węzły nie wymienione w instrukcji **.IC** mają na początku analizy stanu nieustalonego potencjał równy zeru, natomiast pozostałe węzły mają potencjały ustalone przez deklarację **.IC**. Deklaracja nie wpływa na sposób obliczania statycznego punktu pracy za pomocą instrukcji **.OP**.

Patrz strona: 72.

---

**.INC** DOŁĄCZ ZBIÓR

Postać ogólna:

```
.INC _nazwa_zbioru
```

Przykłady:

```
.INC UA741.CIR
```

```
.INC B:\PSPICE\DIODY.LIB
```

Zawartość zbioru o nazwie podanej w polu **\_nazwa\_zbioru** wstawiana jest „mechanicznie” w miejscu występowania instrukcji **.INC**. Wstawiany tekst może zawierać instrukcję **.INC** powodującą wstawianie tekstu na „niższym poziomie”. Dozwolone są cztery poziomy wstawiania tekstu. Instrukcja **.END** oznacza koniec wstawianego tekstu.

Patrz strona: 67.

---

**.LIB****DOŁĄCZ BIBLIOTEKĘ**

Postać ogólna:

```
.LIB [_nazwa_zbioru]
```

Przykłady:

```
.LIB
.LIB OPAMP.LIB
.LIB C:\PSPICE\DIODY.LIB
```

Instrukcja **.LIB** służy w celu odwołania do modelu lub podobwodu umieszczonego w zbiorze dyskowym o nazwie, którą podaje się w polu **\_nazwa\_zbioru**. Zbiór ten może zawierać deklarację modelu (**.MODEL**) deklarację struktury podobwodu (**.SUBCKT**, **.ENDS**) oraz instrukcję **.LIB**. Jeżeli nie zostanie podana nazwa zbioru zakłada się, że chodzi o zbiór **NOM.LIB** leżący w kartotece domyślnej.

Instrukcja **.LIB** (w przeciwieństwie do instrukcji **.INC**) powoduje dołączenie do przetwarzanych danych tylko tych deklaracji, do których następuje odwołanie w definicji struktury obwodu. W ten sposób zmniejsza się zajętość pamięci operacyjnej maszyny cyfrowej.

Patrz strona: 92.

**.MC****ANALIZA MONTE CARLO**

Postać ogólna:

```
.MC _powt [DC][AC][TRAN] _wyjście YMAX [LIST][OUTPUT] _sposób
```

Przykłady:

```
.MC TRAN V(5) YMAX
.MC 50 DC IC(Q7) YMAX LIST
.MC 20 AC VP(13,5) YMAX LIST OUTPUT ALL
```

Analiza Monte Carlo może zostać przeprowadzona dla analizy stałoprądowej — słowo kluczowe **DC** — analizy zmiennoprądowej — słowo kluczowe **AC** — lub analizy stanu nieustalonego — słowo kluczowe **TRAN**. Dokładnie jedno z wymienionych słów kluczowych musi zostać użyte. Odpowiednia analiza zostanie powtórzona tyle razy, ile wskazuje pole **\_powt**. Za każdym razem zmieniane będą parametry modeli opatrzone tolerancjami (słowa kluczowe **DEV** i **LOT** w deklaracji **.MODEL**). Badane będą zmiany wartości wielkości określonej w polu **\_wyjście** (wielkości dopuszczalne przez instrukcję **.PRINT**). Słowo kluczowe **YMAX** (konieczne) oznacza, operację sprowadzającą wynik całej analizy **DC**, **AC** lub **TRAN** do jednej liczby — maksymalnego odchylenia wielkości wyjściowej od jej wartości nominalnej. W chwili obecnej jest to jedyny zaimplementowany typ operacji.

Jeżeli podane zostanie słowo kluczowe **LIST** to program umieści w zbiorze wyjściowym aktualne wartości parametrów modeli dla każdej analizy.

Pierwsza analiza wykonywana jest dla wartości nominalnych parametrów. Wyniki tej analizy podawane są zgodnie z instrukcjami **.PRINT**, **.PLOT**, **.PROBE**. Natomiast wyniki pozostałych analiz są udostępniane na żądanie. Aby je uzyskać po słowie kluczowym **OUTPUT**, w polu **\_sposób**, podaje się jedno z następujących słów kluczowych:

<b>ALL</b>	udostępniane są wyniki wszystkich analiz;
<b>FIRST n</b>	udostępniane są wyniki pierwszych n analiz;

**EVERY n** udostępniane są wyniki co n-tej analizy;  
**RUNS n1 n2 ...** udostępniane są wyniki analizy o numerze n1,n2,...

Patrz strona: 41.

## **.MODEL**

## **DEKLARACJA MODELU**

Postać ogólna:

```
.MODEL _nazwa _typ [<_lista_parametrów>]
```

Przykład:

```
.MODEL RMAX RES R=1.5K TC1=.02 TC2=.005
```

```
.MODEL DNOM D IS=1E-9
```

```
.MODEL QDRIV NPN IS=1E-7 BF=50
```

```
.MODEL MLOAD NMOS LEVEL=1 VTO=0.7 CJ=0.2PF
```

```
.MODEL CMOD CAP C=1 DEV=5%
```

```
.MODEL DLOAD D IS=1E-9 DEV=0.5% LOT=10%
```

Deklaracje elementów odwołują się do modelu poprzez nazwę podaną w polu **\_nazwa**. Nazwa modelu musi zaczynać się literą, przy czym zaleca się aby litera ta była taka sama jak w typie modelu np. D dla diody, Q dla tranzystora bipolarnego itd. Pole **\_typ** identyfikuje rodzaj przyrządu, którego dotyczy model. W polu tym mogą znaleźć się następujące słowa kluczowe:

<b>CAP</b>	kondensator;
<b>IND</b>	cewka;
<b>RES</b>	rezystor;
<b>D</b>	diody;
<b>NPN</b>	tranzystor bipolarny typu n-p-n;
<b>PNP</b>	tranzystor bipolarny typu p-n-p;
<b>NJF</b>	tranzystor polowy złączowy z kanałem typu n;
<b>PJF</b>	tranzystor polowy złączowy z kanałem typu p;
<b>NMOS</b>	tranzystor polowy MOS z kanałem typu n;
<b>PMOS</b>	tranzystor polowy MOS z kanałem typu p;
<b>GASFET</b>	tranzystor polowy złączowy z kanałem typu n wykonany na podłożu GaAs;
<b>CORE</b>	nieliniowy rdzeń magnetyczny;
<b>VSWITCH</b>	klucz sterowany napięciem;
<b>ISWITCH</b>	klucz sterowany prądem.

Deklaracja przyrządu może odwoływać się tylko do modelu przyrządu odpowiedniego typu. Np. deklaracja tranzystora polowego złączowego może odwoływać się do modelu typu **PJF** lub **NJF** ale nie może odwoływać się do modelu typu **NPN**.

Deklaracja modelu kończy się opcjonalną listą parametrów modelu **\_lista\_parametrów**. Jeżeli nie podano wartości żadnego z parametrów modelu przyjęte zostaną wartości domyślne wbudowane w program PSpice.

Dla każdego z parametrów może zostać określona tolerancja po słowie kluczowym **DEV** lub **LOT**:

```
[DEV _wartość [%]][LOT _wartość [%]]
```

Jeżeli tolerancja parametru poprzedzona jest słowem kluczowym **DEV**, to dla każdego przyrządu odwołującego się do danego modelu wartość parametru jest inna. Jeżeli tolerancja

podana jest po słowie kluczowym **LOT**, to dla wszystkich elementów odwołujących się do danego modelu wartość parametru jest taka sama. Tolerancja może być tolerancją względną — określa to znak % — lub tolerancją bezwzględną.

Patrz strony: 21, 42, 111.

### **.NODESET POTENCJAŁY WĘZŁOWE — DEKLARACJA**

Postać ogólna:

```
.NODESET V(_węzeł1)=_w1 [V(_węzeł2)=_w2 ...]
```

Przykład:

```
.NODESET V(2)=3.4 V(102)=0.1 V(3)=-1V
```

Deklaracja `.NODESET` pozwala na podanie wartości początkowych potencjałów węzłowych. Dzięki temu iteracyjne poszukiwanie statycznego punktu pracy układu rozpocząć można od dowolnej wartości potencjałów węzłowych. Może to:

- Ułatwić znalezienie statycznego punktu pracy układu.
- Spowodować, że w wyniku iteracji obliczony zostanie statyczny punkt pracy układu w wybranym stanie — dla układów bistabilnych.

Deklaracja `.NODESET` wpływa na obliczanie statycznego punktu pracy układu instrukcją `.OP` oraz na obliczanie statycznego punktu pracy przed analizą zmiennoprądową oraz analizą stanu nieustalonego.

Patrz strona: 27.

### **.NOISE ANALIZA SZUMÓW**

Postać ogólna:

```
.NOISE _napiecie_wY _wEjście [_dzielnik]
```

Przykłady:

```
.NOISE V(5) VIN
```

```
.NOISE V(101) VSRC 20
```

```
.NOISE V(4,5) IRSC
```

Analiza szumów wykonywana jest podczas analizy zmiennoprądowej układu. Dlatego obok instrukcji `.NOISE` konieczne jest podanie instrukcji `.AC`. Składowe widma mocy szumu pochodzące od każdego z elementów układu są sumowane na wyjściu *napięciowym* określonym w polu **\_napiecie\_wY**. Obliczana jest także gęstość widmowa mocy szumów sprowadzona do wejścia układu określonego w polu **\_wEjście**. W polu tym może znaleźć się:

- Nazwa niezależnego źródła napięcia. Gęstość widmowa mocy szumów sprowadzonych do wejścia ma wtedy wymiar  $[V]/[Hz]^{1/2}$ .
- Nazwa niezależnego źródła prądu. Gęstość widmowa mocy szumów sprowadzonych do wejścia ma wtedy wymiar  $[A]/[Hz]^{1/2}$ .

W opcjonalny polu **\_dzielnik** podaje się liczbę naturalną *n*. Wyniki analizy szumowej będą podawane dla co *n*-tej częstotliwości określonej w instrukcji analizy zmiennoprądowej `.AC`. Jeżeli wspomniana liczba nie wystąpi analiza szumowa przeprowadzona zostanie dla wszystkich częstotliwości (domyślna wartość parametru **\_dzielnik** wynosi 1).

Patrz strona: 65.

**.OP****STATYCZNY PUNKT PRACY**

Postać ogólna:

.OP

Przykład:

.OP

Instrukcja .OP powoduje obliczenie statycznego punktu pracy układu. W zbiorze wyjściowym umieszczone zostaną:

- Wartości potencjałów węzłowych w obwodzie.
- Wartości prądów płynących przez niezależne źródła napięcia.
- Parametry małosygnałowe elementów półprzewodnikowych i nieliniowych źródeł sterowanych.
- Moc rozpraszana przez układ.

Statyczny punkt pracy obliczany jest także w przypadku, gdy zbiór danych wejściowych zawiera jedynie opis struktury obwodu. Jednak w zbiorze wyjściowym umieszczone zostaną tylko wartości potencjałów węzłowych.

Patrz strona: 19.

**.OPTIONS****PARAMETRY STERUJĄCE**

Postać ogólna:

.OPTIONS <\_lista\_opcji>

Przykłady:

.OPTIONS NOECHO NOMOD DEFL=12U DEFW=8U DEFAD=150P

.OPTIONS ACCT RELTOL=0.01

Parametry sterujące działaniem programu PSpice, umieszczane na liście **\_lista\_opcji**, można podzielić na dwie grupy. Pierwsza grupa parametrów to flagi. Na liście wystarczy umieścić słowo kluczowe identyfikujące parametr. Poniżej podano wspomniane słowa kluczowe z wyjaśnieniem ich znaczenia.

<b>ACCT</b>	W zbiorze wyjściowym umieszczone zostaną dodatkowe dane dotyczące czasu obliczeń, liczby równań, liczby elementów, zajętości pamięci itd.
<b>LIST</b>	W zbiorze wyjściowym umieszczane jest zestawienie elementów obwodu i ich parametrów.
<b>NODE</b>	W zbiorze wyjściowym umieszcza się listę zawierającą poszczególne węzły obwodu i dołączone do nich elementy.
<b>NOECHO</b>	Powoduje, że w zbiorze wyjściowym nie zostanie powtórzony opis struktury obwodu.
<b>NOMOD</b>	Powoduje, że w zbiorze wyjściowym nie będą umieszczane parametry zadeklarowanych modeli ani parametry modeli przeliczone dla temperatur innych niż nominalna.
<b>NOPAGE</b>	W zbiorze wyjściowym nie będą umieszczane znaki końca strony.
<b>OPTS</b>	W zbiorze wyjściowym umieszczone zostaną wartości wszystkich opcji.
<b>WIDTH</b>	Oznacza to samo co instrukcja .WIDTH OUT=.

Do drugiej grupy parametrów należą te parametry, którym przypisuje się wartości. Po słowie

kluczowym identyfikującym parametr podaje się wartość parametru. Separatorem jest znak „=” (przykłady powyżej). Poniżej przedstawiono słowa kluczowe identyfikujące poszczególne parametry.

<b>ABSTOL</b>	Bezwzględna wartość błędu dla prądów. Wartość domyślna: 1[pA].
<b>CHGTOL</b>	Bezwzględna wartość błędu dla ładunków. Wartość domyślna: 0.01[pC].
<b>CPTIME</b>	Dopuszczalny czas pracy jednostki centralnej komputera. Wartość domyślna: 10 <sup>6</sup> [s]
<b>DEFAD</b>	Wartość domyślna pola powierzchni obszaru dyfuzji drenu dla tranzystorów MOS. Wartość domyślna: 0.0[m] <sup>2</sup> .
<b>DEFAS</b>	Wartość domyślna pola powierzchni obszaru dyfuzji źródła dla tranzystorów MOS. Wartość domyślna: 0.0[m] <sup>2</sup> .
<b>DEFL</b>	Wartość domyślna długości kanału tranzystora MOS. Wartość domyślna: 100.0[μm].
<b>DEFW</b>	Wartość domyślna szerokości kanału tranzystora MOS. Wartość domyślna: 100.0[μm].
<b>GMIN</b>	Minimalna wartość konduktancji gałęzi. Wartość domyślna: 10 <sup>-12</sup> [Ω] <sup>-1</sup> .
<b>ITL1</b>	Największa liczba iteracji podczas obliczania statycznego punktu pracy układu. Wartość domyślna: 40.
<b>ITL2</b>	Największa liczba iteracji podczas obliczania charakterystyk stałoprądowych. Wartość domyślna: 20.
<b>ITL4</b>	Największa liczba iteracji podczas obliczania stanu nieustalonego w wybranej chwili czasu. Wartość domyślna: 10.
<b>ITL5</b>	Największa liczba wszystkich iteracji podczas obliczania stanu nieustalonego (ITL5=0 oznacza, że ITL5=∞). Wartość domyślna: 5000.
<b>LIMPTS</b>	Największa liczba punktów (częstotliwości, chwil czasu itp.), które można umieścić w tabeli lub na wykresie (dotyczy tylko instrukcji .PRINT i .PLOT). Wartość domyślna: 201.
<b>NUMDGT</b>	Liczba cyfr drukowanych liczb (umieszczanych w zbiorze wyjściowym). Maksymalna liczba cyfr równa jest 8. Wartość domyślna: 4.
<b>PIVREL</b>	Względna wartość wymagana dla elementów macierzy admitancyjnej leżących na przekątnej głównej. Wartość domyślna: 10 <sup>-3</sup> .

<b>PIVTOL</b>	Bezwzględna wartość wymagana dla elementów macierzy admitancyjnej leżących na przekątnej głównej. Wartość domyślna: $10^{-13}$ .
<b>RELTOL</b>	Dokładność względna wymagana dla napięć i prądów. Wartość domyślna: 0.001.
<b>TNOM</b>	Nominalna temperatura otoczenia analizowanego układu. Wartość domyślna: 27°C.
<b>TRTOL</b>	Współczynnik korekcji dla dopuszczalnej wartości lokalnego błędu obciążenia. Wartość domyślna: 7.0.
<b>VNTOL</b>	Bezwzględna wartość błędu dla napięć. Wartość domyślna: 1.0[ $\mu$ V].

Patrz strony: 24, 85, 130, 135, 166.

**.PLOT****WYKRES**

Postać ogólna:

```
.PLOT [DC][AC][NOISE][TRAN] _wY1 [_lo1,_hi1] _wY2 [_lo2,_hi2] ...
```

Przykłady:

```
.PLOT DC V(3) V(R1), V(2,3) I(VIN) I(R2) IB(Q13) VBE(Q12)
```

```
.PLOT AC VM(2) VM(2,4) VG(5) VDB(5) IR(6) II(7)
```

```
.PLOT NOISE INOISE ONOISE DB(INOISE) DB(ONOISE)
```

```
.PLOT TRAN V(3) V(3,2) 0,5V ID(M2) I(VCC) (-50mA,50mA)
```

W zbiorze wyjściowym tworzony jest wykres wielkości określonych w polach **\_wY1**, **\_wY2**,... Dla każdej wielkości można określić w polu **\_lo1** (**\_lo2**, ...) najmniejszą wartość umieszczoną na osi pionowej oraz w polu **\_hi1** (**\_hi2**, ...) największą wartość umieszczoną na osi pionowej. Wykres dotyczy tylko jednego typu analizy, który musi być określony przez podanie dokładnie jednego spośród słów kluczowych **DC**, **AC**, **NOISE**, **TRAN**. Wykres tworzony jest za pomocą standardowych znaków pisarskich tak, że może być wydrukowany na każdym typie drukarki.

Patrz strona: 55.

**.PRINT****TABLICA**

Postać ogólna:

```
.PRINT [DC][AC][NOISE][TRAN] <_lista_wielkości>
```

Przykłady:

```
.PRINT DC V(3) V(R1), V(2,3) I(VIN) I(R2) IB(Q13) VBE(Q12)
```

```
.PRINT AC VM(2) VM(2,4) VG(5) VDB(5) IR(6) II(7)
```

```
.PRINT NOISE INOISE ONOISE DB(INOISE) DB(ONOISE)
```

```
.PRINT TRAN V(3) V(3,2) ID(M2) I(VCC)
```

W zbiorze wyjściowym tworzona jest tablica wartości dla wielkości umieszczonych na liście **\_lista\_wielkości**. Instrukcja **.PRINT** dotyczy tylko jednego typu analizy, który musi być określony przez podanie dokładnie jednego spośród słów kluczowych **DC**, **AC**, **NOISE**, **TRAN**.

Patrz strona: 52.

**.PROBE****PROGRAM GRAFICZNY**

Postać ogólna:

```
.PROBE [<_lista_wielkości>]
```

Przykłady:

```
.PROBE
```

```
.PROBE V(3) V(3,2) ID(M2) I(VCC) I(R2) IB(Q13) VBE(Q12)
```

Wyniki wszystkich analiz (**DC,AC,NOISE,TRAN**) umieszczone zostają w zbiorze dyskowym o nazwie PROBE.DAT i w ten sposób przekazane procesorowi graficznemu Probe. Przekazywane są wielkości wyszczególnione na liście **\_lista\_wielkości**. Jeżeli instrukcja .PROBE wystąpi bez parametrów, to do programu Probe przekazane zostaną wszystkie możliwe przebiegi.

Patrz strony: 22, 35, 56, 65.

**.SENS****ANALIZA WRAŻLIWOŚCI**

Postać ogólna:

```
.SENS <_lista_wielkości>
```

Przykład:

```
.SENS V(9) V(3,4) V(17) I(VCC)
```

Instrukcja powoduje obliczenie stałoprądowych, różniczkowych wrażliwości wielkości, które wyszczególniono na liście **\_lista\_wielkości**. Wrażliwości obliczane są względem zmian wszystkich parametrów w obwodzie. Zwykle prowadzi to do wygenerowania olbrzymiego zbioru wyjściowego, zawierającego wyniki analizy. Lista wielkości będąca parametrem instrukcji .SENS może zawierać wielkości dopuszczalne przez instrukcję .PRINT dla analizy stałoprądowej.

Patrz strona: 36.

**.SUBCKT****DEKLARACJA PODOBWODU**

Postać ogólna:

```
.SUBCKT _nazwa <_lista_węzłów>
```

Przykład:

```
.SUBCKT WZMACNIACZ 1 2 101 102
```

Deklaracja podobwodu zaczyna się od deklaracji .SUBCKT, a kończy deklaracją .ENDS. Struktura podobwodu deklarowana jest w liniach leżących między wymienionymi dwoma deklaracjami. Na opis struktury podobwodu mogą składać się:

- Deklaracje elementów.
- Deklaracje modeli matematycznych przyrządów — .MODEL.
- Odwołania do podobwodów — pseudo-element typu X.

*Wewnątrz podobwodu nie można definiować innego podobwodu.* Wszystkie nazwy i numery węzłów zdefiniowane wewnątrz podobwodu mają znaczenie lokalne i nie są rozpoznawane poza podobwodem (powtórnie można użyć tych samych nazw). Wszystkie podobwody i modele zdefiniowane poza danym podobwodem są w tym podobwodzie dostępne. Podobwód

wywoływany jest przez pseudoelement, którego nazwa zaczyna się od litery „X”. Odwołanie następuje przez nazwę określoną w polu **\_nazwa**. Na liście **\_lista\_węzłów** podane są wewnętrzne numery węzłów, które udostępniane są na zewnątrz podobwołu. Deklaracja pseudoelementu zawiera numery węzłów obwołu, do których dołączone zostaną udostępniane na zewnątrz węzły podobwołu.

Patrz strona: 68.

---

**.TEMP****TEMPERATURA**

Postać ogólna:

```
.TEMP <_lista_temperatur>
```

Przykład:

```
.TEMP -10 20 30 70
```

Instrukcja powoduje wykonanie każdej zleconej analizy dla każdej temperatury otoczenia analizowanego obwołu, która wymieniona została na liście **\_lista\_temperatur**. Zakłada się, że parametry modeli dotyczą temperatury nominalnej TNOM (deklaracja .OPTIONS). Wartość domyślna parametru TNOM wynosi 27°C. Temperatury podane w instrukcji .TEMP powinny być wyrażone w °C. Temperatury niższe niż -273°C są ignorowane.

Patrz strona: 135.

---

**.TF****TRANSMITANCJA**

Postać ogólna:

```
.TF _wYjście _źródło
```

Przykłady:

```
.TF V(5) VIN
```

```
.TF I(VDRIV) ICVT
```

Obliczana jest małosygnalowa transmitancja od wymuszenia (niezależnego źródła napięcia lub prądu) określonego w polu **\_źródło** do wielkości wyjściowej (dowolne napięcie lub prąd płynący przez niezależne źródło napięcia) określonej w polu **\_wYjście**. Wynik obliczeń automatycznie umieszczany jest w zbiorze z danymi wyjściowymi.

Patrz strona: 33.

---

**.TRAN****ANALIZA STANU NIEUSTALONEGO**

Postać ogólna:

```
.TRAN[/OP] _d_krok _stop [_nie_druk [_krok]] [[UIC]]
```

Przykłady:

```
.TRAN 1ns 100ns
```

```
.TRAN/OP 1ns 100ns 20ns UIC
```

```
.TRAN 1ns 100ns 0ns 0.1ns
```

Instrukcja powoduje wykonanie analizy stanów nieustalonych w przedziale czasu od t=0 do t=**\_stop**. Krok czasowy używany w przypadku, gdy wyniki analizy wyprowadzane są za pomocą instrukcji .PRINT lub instrukcji .PLOT określony jest w polu **\_d\_krok**. Program PSpice oblicza przebiegi czasowe stosując zmienny krok całkowania tak, aby zachować stałą wartość lokalnego błędu obciążenia. Ponieważ wyprowadzane wartości przebiegów powinny być

równomiernie spróbkowane program dokonuje interpolacji trójmianem kwadratowym.

Analiza wykonywana jest zawsze od chwili  $t=0$ . Natomiast w polu **\_nie\_druk** może być określona chwila czasu od której zacznie się wyprowadzanie wyników analizy. W ten sposób pomija się wyniki z przedziału czasu od  $t=0$  do  $t=$ **\_nie\_druk**.

Wartość kroku całkowania stosowanego „wewnętrznie” przez program PSpice jest ograniczona od góry do wartości **\_stop/50.0**. Jeżeli ograniczenie to chcemy zmienić, to maksymalną wartość kroku całkowania można podać w polu **\_krok**.

Przed analizą stanu nieustalonego obliczany jest statyczny punkt pracy układu. Jeżeli użyjemy przyrostka **/OP**, to tak jak w przypadku instrukcji **.OP** w zbiorze wyjściowym umieszczone zostaną informacje na temat potencjałów węzłowych, prądów płynących przez niezależne źródła napięcia, parametrów małosygnalowych elementów nieliniowych itd. Jeżeli chcemy pominąć obliczanie statycznego punktu pracy i przyjąć warunki początkowe określone przez deklarację **.IC** lub deklaracje **IC=** w liniach deklaracji poszczególnych elementów należy użyć słowa kluczowego **UIC**.

Patrz strony: 72, 98.

---

### **.WIDTH** DŁUGOŚĆ LINII ZBIORU WYJŚCIOWEGO

Postać ogólna:

```
.WIDTH OUT=_wartość
```

Przykłady:

```
.WIDTH OUT=80
```

```
.WIDTH OUT=132
```

Instrukcja zmienia długość linii zbioru wyjściowego, wyrażoną przez liczbę kolumn. Dopuszczalna liczba kolumn wynosi 80 lub 132.

Patrz strona: 5.